load($$[STARLAB])
load($$[SURFACEMESH])
load($$[EIGEN])

TEMPLATE = lib
CONFIG += staticlib

QT += opengl

# Build flag
CONFIG(debug, debug|release) {
    CFG = debug
} else {
    CFG = release
}

# Library name and destination
TARGET = SymDetectorLib
DESTDIR = $$PWD/$$CFG/lib

# MeanShift library
LIBS += -L$$PWD/../MeanShift/$$CFG/lib -lMeanShift
INCLUDEPATH += ../MeanShift

# ICP library
LIBS += -L$$PWD/../ICP/$$CFG/lib -lICP
INCLUDEPATH += ../ICP

# Global headers
INCLUDEPATH *= $$PWD/../Global
DEPENDPATH  *= $$PWD/../Global

# Dirty access to sym-detect_plugin
INCLUDEPATH *= $$PWD/../sym-detect
DEPENDPATH  *= $$PWD/../sym-detect

HEADERS += \
    SymmetryVoter.h \
    SymmetryDetector.h \
    Sampler.h \
    PointSignature.h \
    Monge_via_jet_fitting.h

SOURCES += \
    SymmetryVoter.cpp \
    SymmetryDetector.cpp \
    Sampler.cpp \
    PointSignature.cpp \
    Monge_via_jet_fitting.cpp
